

# **Microelectronic Circuits International 8<sup>th</sup> Edition**


A. Sedra, K.C. Smith  
T. Chan Carusone, V. Gaudet

*Spice Problems Solutions  
Chapter 12*

*Prepared by: Nijwm Wary  
2019*

**Problem: 12.49**

- The schematics for part (a) this problem is shown below



- Run the netlist and perform operating point analysis. The node voltages are

| NODE    | VOLTAGE | NODE     | VOLTAGE | NODE     | VOLTAGE | NODE     | VOLTAGE  |
|---------|---------|----------|---------|----------|---------|----------|----------|
| ( VO )  | .0154   | ( VB7 )  | .6805   | ( VCC )  | 10.0000 | ( VEE )  | -10.0000 |
| ( VIN ) | 0.0000  | ( VC12 ) | -9.4223 | ( VC24 ) | -9.3346 | ( VE12 ) | .6346    |


You will notice there is an offset present in the simulation, which was not evident in the hand calculation. Hence, the negative opamp input terminal, fed back from the output, is 15 mV above ground.

- The currents are,


|       |          |           |          |           |          |
|-------|----------|-----------|----------|-----------|----------|
| NAME  | Q3       | Q1        | Q4       | Q2        | Q6       |
| MODEL | Q2N3904  | Q2N3906   | Q2N3904  | Q2N3906   | Q2N3904  |
| IB    | 6.80E-07 | -3.52E-07 | 6.80E-07 | -6.38E-07 | 1.50E-05 |
| IC    | 3.38E-05 | -3.52E-05 | 3.38E-05 | -6.38E-05 | 9.85E-04 |
| NAME  | Q5       | Q7        |          |           |          |
| MODEL | Q2N3904  | Q2N3904   |          |           |          |
| IB    | 1.50E-05 | 1.48E-05  |          |           |          |
| IC    | 9.85E-04 | 9.73E-04  |          |           |          |

Again, the opamp offset is evident as the collector currents in Q1 and Q2 are not precisely matched.


4. The schematic for part (b), (c) and (d) is shown below. Note that in this case, the op-amp offset (-15.43mV) is added in input to make output zero.




5. Perform AC analysis and plot  $(V(IN+) - V(IN-)) / (I(V1))$  to calculate the differential input resistance.



6. The gain of the amplifier is calculated by plotting the expression  $DB(V(VO) / (V(IN+) - V(IN-)))$ . Note that



7. For part (c), sweep the  $C_c$  from 0.1nF to 0.3nF and find the capacitance where the bandwidth is 100Hz. The required capacitance is somewhat larger than predicted by hand calculation because the opamp dc gain is lower than calculated.



**Netlist:**

Copy the netlist given below for part (a) and paste it into a text file and save it with \*.cir extension.

```

*****Problem: P13_66 (a) *****
***** Main circuit begins here*****
V_CC          VCC 0 10Vdc
Q3            VC12 VC12 VEE Q2N3904
Q1            VC12 VO VE12 Q2N3906
RL            0 VO 5k
I1            VCC VE12 DC 0.1mAdc
Q4            VC24 VC12 VEE Q2N3904
C             VC24 VB7 {Cc}
Q2            VC24 VIN VE12 Q2N3906
Q6            VO VC24 VEE Q2N3904
Q5            VB7 VC24 VEE Q2N3904
I2            VCC VB7 DC 1mAdc
Q7            VCC VB7 VO Q2N3904
V_EE          0 VEE 10Vdc
V1            VIN 0 AC 10m
+SIN 0 10m 1k 0 0 0
.PARAM cc=10n
***** Main circuit ends here*****

***** Q2N3906 model begins here *****
.model Q2N3906 PNP(Is=1.41f Xti=3 Eg=1.11 Vaf=10000 Bf=100 Ne=1.5 Ise=0
+           Ikf=80m Xtb=1.5 Br=4.977 Nc=2 Isc=0 Ikr=0 Rc=2.5 Cjc=9.728p
+           Mjc=.5776 Vjc=.7 Fc=.5 Cje=8.063p Mje=.3677 Vje=.7 Tr=33.42n
+           Tf=179.3p Itf=.4 Vtf=4 Xtf=6 Rb=10)
***** Q2N3906 model ends here *****

***** Q2N3904 model begins here *****
.model Q2N3904 NPN(Is=6.734f Xti=3 Eg=1.11 Vaf=10000 Bf=100 Ne=1.259
+           Ise=6.734f Ikf=66.78m Xtb=1.5 Br=.7371 Nc=2 Isc=0 Ikr=0 Rc=1
+           Cjc=3.638p Mjc=.3085 Vjc=.7 Fc=.5 Cje=4.493p Mje=.2593 Vje=.7
+           Tr=239.5n Tf=301.2p Itf=.4 Vtf=4 Xtf=2 Rb=10)
***** Q2N3904 model ends here *****

***** Analysis begins here*****
.OP
***** Analysis ends here*****

```

Copy the netlist given below for part (b), (c) and (d), and paste it into a text file and save it with \*.cir extension.

```

*****Problem: P13_66 (b), (c), (d) *****
***** Main circuit begins here*****
V_CC          VCC 0 10Vdc
Q3            VC12 VC12 VEE Q2N3904
Q1            VC12 IN- VE12 Q2N3906
RL            0 VO {RL}
I1            VCC VE12 DC 0.1mAdc
Q4            VC24 VC12 VEE Q2N3904
C             VC24 VB7 {Cc}
Q2            VC24 IN+ VE12 Q2N3906
Q6            VO VC24 VEE Q2N3904
Q5            VB7 VC24 VEE Q2N3904
I2            VCC VB7 DC 1mAdc
Q7            VCC VB7 VO Q2N3904
V_EE          0 VEE 10Vdc
V1            IN+ 0 AC 1
+SIN -15.43m 10m 1k 0 0 0
V2            IN- 0 AC -1
+SIN 0 10m 1k 0 0 0
.PARAM cc=10n rl=5k
***** Main circuit ends here*****

```

```

***** Q2N3906 model begins here *****
.model Q2N3906 PNP(Is=1.41f Xti=3 Eg=1.11 Vaf=10000 Bf=100 Ne=1.5 Ise=0
+           Ikf=80m Xtb=1.5 Br=4.977 Nc=2 Isc=0 Ikr=0 Rc=2.5 Cjc=9.728p
+           Mjc=.5776 Vjc=.7 Fc=.5 Cje=8.063p Mje=.3677 Vje=.7 Tr=33.42n
+           Tf=179.3p Itf=.4 Vtf=4 Xtf=6 Rb=10)
***** Q2N3906 model ends here *****

***** Q2N3904 model begins here *****
.model Q2N3904 NPN(Is=6.734f Xti=3 Eg=1.11 Vaf=10000 Bf=100 Ne=1.259
+           Ise=6.734f Ikf=66.78m Xtb=1.5 Br=.7371 Nc=2 Isc=0 Ikr=0 Rc=1
+           Cjc=3.638p Mjc=.3085 Vjc=.7 Fc=.5 Cje=4.493p Mje=.2593 Vje=.7
+           Tr=239.5n Tf=301.2p Itf=.4 Vtf=4 Xtf=2 Rb=10)
***** Q2N3904 model ends here *****
***** Analysis begins here*****
.AC DEC 20 1 1K
*.STEP LIN PARAM cc 0.1n 0.3n 0.025n
.PROBE
.END
***** Analysis ends here*****

```